
(* Base problem name and configuration *)

myFile = "SSF-082_1kHz.fem";

myCoilName = "Icoil";

myCoilGroup = 1;

(* Displacement range under consideration *)

xmin = -15;

xmax = 15;

dx = 2;

(* Frequency range under consideration in Hz *)

w = 1000;

<< c:\\femm42\\mathfemm\\mathfemm.m

MathFEMM loaded at Tue 3 Nov 2015 21:56:55

OpenFEMM[]

(* Open up base problem and save as the DC operating point with no coil current *)

OpenDocument[NotebookDirectory[] <> myFile];

MISetCurrent[myCoilName, 0];

MISaveAs[NotebookDirectory[] <> "DCProblem.fem"];

(* Open up base problem and save as the

incremental AC problem with a coil current of 1A *)

(* Incremental AC problem points back to the DC solution

via the MISetPrevious["DCProblem.ans",1] call *)

OpenDocument[NotebookDirectory[] <> myFile];

MISetCurrent[myCoilName, 1];

MIProbDef[w];

MISetPrevious["DCProblem.ans", 1];

MISaveAs[NotebookDirectory[] <> "ACProblem.fem"];

(* Move the coil to the xmin position,

assuming that the geometry is drawn so that the coil is nominally at x=0 *)

DCSetFocus[] := MISetFocus["DCProblem.fem"];

ACSetFocus[] := MISetFocus["ACProblem.fem"];



DCSetFocus[];

MISelectGroup[myCoilGroup];

MIMoveTranslate[0, xmin];

ACSetFocus[];

MISelectGroup[myCoilGroup];

MIMoveTranslate[0, xmin];

(* Step through the full range of coil positions *)

Lx = {};

Vx = {};

Bl = {};

(* At each coil position,

evaluate Bl and coil inductance over a range of frequencies *)

For[x = xmin, x ≤ xmax, x += dx,

DCSetFocus[];

MIAnalyze[];

ACSetFocus[];

MIAnalyze[];

MILoadSolution[];

(* Get Bl *)

MOGroupSelectBlock[1];

f = Abs[MOBlockIntegral[29]];

Bl = Append[Bl, {x, f}];

(* Get Inductance and Voltage *)

u = MOGetCircuitProperties[myCoilName];

Lx = Append[Lx, {x, u[[3]]}];

Vx = Append[Vx, {x, u[[2]]}];

Print[{x, u[[3]]}];

If[x < xmax,

DCSetFocus[];

MISelectGroup[myCoilGroup];

MIMoveTranslate[0, dx];

ACSetFocus[];

MISelectGroup[myCoilGroup];

MIMoveTranslate[0, dx];

];

]

{-15, 0.00482814- 0.00278792 ⅈ}

{-13, 0.00496937- 0.00270284 ⅈ}

{-11, 0.00511339- 0.00259933 ⅈ}

{-9, 0.00522405- 0.00246864 ⅈ}

{-7, 0.00526879- 0.0023075 ⅈ}

{-5, 0.00524326- 0.00212125 ⅈ}

{-3, 0.00516112- 0.00191732 ⅈ}

{-1, 0.00504053- 0.00170831 ⅈ}

{1, 0.00491105- 0.0015036 ⅈ}

{3, 0.00477723- 0.00130941 ⅈ}

{5, 0.00462044- 0.0011242 ⅈ}

{7, 0.0044263- 0.000951303 ⅈ}

{9, 0.00419797- 0.000786545 ⅈ}

{11, 0.00393325- 0.000634708 ⅈ}

{13, 0.00365867- 0.000499021 ⅈ}

{15, 0.00341709- 0.000384987 ⅈ}

CloseFEMM[]

ListPlot[Transpose[{Vx.{1, 0}, Abs[Vx.{0, 1}]}], Joined → True, Frame → True,

GridLines → Automatic, ImageSize → 500, BaseStyle → {FontFamily → "Tahoma", FontSize → 14},

FrameLabel → {"Displacement, mm", "Impedance, Ohms"}]

ListPlot[Transpose[{Lx.{1, 0}, Abs[Lx.{0, 1}]}], Joined → True, Frame → True,

GridLines → Automatic, ImageSize → 500, BaseStyle → {FontFamily → "Tahoma", FontSize → 14},

FrameLabel → {"Displacement, mm", "|Inductance|, Henries"}]

ListPlot[Re[Lx], Joined → True, Frame → True, GridLines → Automatic,

ImageSize → 500, BaseStyle → {FontFamily → "Tahoma", FontSize → 14},

FrameLabel → {"Displacement, mm", "Re[Inductance], Henries"}]

ListPlot[Re[Lx.{{1, 0}, {0, I * 2 * Pi * w}}], Joined → True, Frame → True,

GridLines → Automatic, ImageSize → 500, BaseStyle → {FontFamily → "Tahoma", FontSize → 14},

FrameLabel → {"Displacement, mm", "Re[I*ω*Inductance], Ohms"}]

Plot[Interpolation[Bl][X], {X, Min[Bl.{1, 0}], Max[Bl.{1, 0}]}, Frame → True,

GridLines → Automatic, ImageSize → 500, BaseStyle → {FontFamily → "Tahoma", FontSize → 14},

FrameLabel → {"Coil Displacement, mm", "Bl, Newtons/Ampere"}]

Lx

{{-15, 0.00482814- 0.00278792 ⅈ}, {-13, 0.00496937- 0.00270284 ⅈ},

{-11, 0.00511339- 0.00259933 ⅈ}, {-9, 0.00522405- 0.00246864 ⅈ},

{-7, 0.00526879- 0.0023075 ⅈ}, {-5, 0.00524326- 0.00212125 ⅈ},

{-3, 0.00516112- 0.00191732 ⅈ}, {-1, 0.00504053- 0.00170831 ⅈ},

{1, 0.00491105- 0.0015036 ⅈ}, {3, 0.00477723- 0.00130941 ⅈ},

{5, 0.00462044- 0.0011242 ⅈ}, {7, 0.0044263- 0.000951303 ⅈ},

{9, 0.00419797- 0.000786545 ⅈ}, {11, 0.00393325- 0.000634708 ⅈ},

{13, 0.00365867- 0.000499021 ⅈ}, {15, 0.00341709- 0.000384987 ⅈ}}

Vx

{{-15, 23.283+ 30.3361 ⅈ}, {-13, 22.7484+ 31.2235 ⅈ},

{-11, 22.098+ 32.1284 ⅈ}, {-9, 21.2769+ 32.8237 ⅈ},

{-7, 20.2644+ 33.1048 ⅈ}, {-5, 19.0942+ 32.9444 ⅈ}, {-3, 17.8129+ 32.4283 ⅈ},

{-1, 16.4996+ 31.6706 ⅈ}, {1, 15.2134+ 30.8571 ⅈ}, {3, 13.9932+ 30.0162 ⅈ},

{5, 12.8295+ 29.0311 ⅈ}, {7, 11.7432+ 27.8113 ⅈ}, {9, 10.708+ 26.3766 ⅈ},

{11, 9.75396+ 24.7134 ⅈ}, {13, 8.90142+ 22.9881 ⅈ}, {15, 8.18492+ 21.4702 ⅈ}}

Bl

{{-15, 6.96031}, {-13, 8.85003}, {-11, 10.7238}, {-9, 12.5628}, {-7, 14.1039},

{-5, 15.1344}, {-3, 15.7471}, {-1, 15.9635}, {1, 15.9487}, {3, 15.7279},

{5, 15.1637}, {7, 14.2276}, {9, 12.7492}, {11, 11.0048}, {13, 9.22781}, {15, 7.44816}}

